(LISP Bulletin) (by D. G. Bobrow)

NO. 1

This first (long delayed) LISP Bulletin contains samples of most of those types of items which the editor feels are relevant to this publication. These include announcements of new (i.e. not previously announced here) implementations of LISP (or closely related) systems; quick tricks in LISP; abstracts of LISP related papers; short writeups and listings of useful programs; and longer articles on problems of general interest to the entire LISP community. Printing of these last articles in the Bulletin does not interfere with later publications in formal journals or books. Short write-ups of new features added to LISP are of interest, preferably upward compatible with LISP 1.5, especially if they are illustrated by programming examples.

A NEW LISP FEATURE

Bobrow, Daniel G., Bolt Beranek and Newman Inc., 50 Moulton Street, Cambridge, Massachusetts 02138.

An extension of prog2, called progn is very useful. The value of progn[e_1;e_2; ...;e_n] is the value of e_n. In BBN-LISP on the SDS 940 we have extended cond to include an implicit progn in each clause, putting it in the general form

$$(\text{COND } (e_{11} \cdots e_{1n_1}) (e_{21} \cdots e_{2n_2}) \cdots (e_{k1} \cdots e_{kn_k}))$$

where $n_1 \geq 1$. This form is identical to the LISP 1.5 form if $n_i = 2$. If $n_1 > 2$ then each expression e_{1k} in a clause is evaluated (in order) when e_{11} is the first true (non-NIL) predicate found. The value of the cond is the value of the last clause evaluated. This is directly extrapolated to the case where $n_1 = 1$, where the value of the cond is the value of this first non-NIL predicate. As an example of the use of this consider

```
(SETQ Z
     (COND ((CDR X))
           (T (SETQ X (COPY A))
              (CONS Y X))))
```

which sets z to cdr[x] if cdr[x] is not NIL (without recomputing the value as would be necessary in LISP 1.5). If cdr[x] is NIL, then x is set and the value of z is set to cons[y;x]. This form of cond is also used in LISP 1.6 for the PDP-6/10; it has proven very convenient, and has no apparent drawbacks.

(LB1, p2)

MLISP USERS' MANUAL

Smith, David Canfield. Stanford Artificial Intelligence Project Memo AI-84. 1969 January.

MLISP is a LISP pre-processor designed to facilitate the writing, use, and understanding of LISP programs. This is accomplished through parentheses reduction, comments, introduction of a more visual flow of control with block structure and mnemonic key words, and language redundancy. In addition, some "meta-constructs" are introduced to increase the power of the language. (Abstract)

The MLISP pre-processor was written by Horace Enea for the IBM 360/67. The author has implemented MLISP on the PDP-6/10 and has added a few auxiliary features, among them compiled object programs, floating point numbers, a new FOR loop, improved error messages and recovery, and additional string manipulation facilities.

TOWARD A PROGRAMMING LABORATORY

Teitelman, Warren. Bolt Beranek and Newman Inc. (International Joint Conference on Artificial Intelligence). 1969 April.

This paper discusses the feasibility and desirability of constructing a "programming laboratory" which would cooperate with the user in the development of his programs, freeing him to concentrate more fully on the conceptual difficulties of the problem he wishes to solve. Experience with similar systems in other fields indicates that such a system would significantly increase the programmer's productivity.

The PILOT system, implemented within the interactive BBN-LISP system, is a step in the direction of a programming laboratory. PILOT operates as an interface between the user and his programs, monitoring both the requests of the user and the operation of his programs. For example, if PILOT detects an error during the execution of a program, it takes the appropriate corrective action based on previous instructions from the user. Similarly, the user can give directions to PILOT about the operation of his programs, even while they are running, and PILOT will perform the work required. In addition, the user can easily modify PILOT by instructing it about its own operation, and thus develop his own language and conventions for interacting with PILOT.

Several examples are presented. (Abstract)

(LB1, p3)

LISP 1.5 Systems

The following are LISP systems the editor has been told about. Please send information about other systems - including write-ups and manuals.

Machine	OS	Location
IBM 360/50	ADEPT	System Development Corporation Santa Monica, California
IBM 360/50	OS/DOS	Rensselaer Polytechnic Institute Troy, New York
IBM/360/91	OS	IBM - Yorktown, New York
PDP 6/10	ITS	Massachusetts Institute of Technology Cambridge, Massachusetts
	10/50	Stanford University, Stanford, California Bolt Beranek and Newman Inc., Cambridge, Massachusetts
CDC 3300		University of Waterloo, Ontario, Canada
CDC 6600		University of Texas, Austin Texas
SDS 940	BBNTS	Bolt Beranek and Newman Inc. Cambridge, Massachusetts

(LB1, p4)

COMMENTING LISP PROGRAMS

Shaw, Christopher J. System Development Corporation. 2500 Colorado Avenue. Santa Monica, California 90406.

Many observers have commented on the lack of comments in LISP programs, and it is truly unfortunate that the designers of LISP never saw fit to incorporate in their systems a capability for disregarding comments.

Though some LISP programmers may disdain them, comments are an indispensible part of any code that is meant to be easily comprehended by humans. And there is really no good excuse for doing without them in LISP; a comment purging process is easily prefixed to most any LISP processor, and if that processor has a macro capability, the job is almost trivial, as shown by the following macro definition:

```
MACRO
(((NOTED
   (LAMBDA (FORM)
    ((LABEL PURGE
      (LAMBDA (X)
       (COND
        ((ATOM X) X)
        ((ATOM (CAR X)) (CONS (CAR X) (PURGE (CDR X))))
        ((EQ (CAAR X) (QUOTE NOTE)) (PURGE (CDR X)))
        (T (LIST (PURGE (CAR X)) (PURGE (CDR X))))
    )))
    (CDR FORM)
)))))
```

This macro operates on any list beginning with the atom NOTED--ordinarily, this would be the list following the DEFINE--by eliminating NOTED and purging any sublist beginning with the atom NOTE. For example:

```
DEFINE
(NOTED
 (NOTE THAT THIS LIST GETS PURGED)
 :::
)
```

would be replaced by

```
DEFINE
(
:::
)
```

(LB1, p5)

A PROGRAM TO DOCUMENT LISP PROGRAM STRUCTURE

Bobrow, Daniel G. Bolt Beranek and Newman Inc. Cambridge, Massachusetts. 1968 January.

In trying to work with large programs, a user can lose track of the hierarchy which defines his program structure; it is often convenient to have a map to show which functions are called by each of the functions in a system. The program PRINTSTRUCTURE, and its six auxiliary functions PRINTSTRUC, PROGSTRUC, ALLCALLS, PRGSTRC, PRGSTRC1 and NOTFN provide an aid for the documentation of this aspect of a system. If FN is the name of the top level function called in your system, then typing in

PRINTSTRUCTURE(FN)

will cause a tree printout of the function-call structure of FN. To describe this in more detail we use the PRINTSTRUCTURE program itself as an example. (its listing is appended to this document) For this function, we get:

```
*PRINTSTRUCTURE(PRINTSTRUCTURE)

PRINTSTRUCTURE  PRINTSTRUC  PRINTSTRUC
                PROGSTRUC   PRGSTRC   NOTFN
                                      PROGSTRUC
                                      PRGSTRC1   PRGSTRC
                                      PRGSTRC
                ALLCALLS    ALLCALLS

(PRINTSTRUCTURE PRINTSTRUC PROGSTRUC PRGSTRC NOTFN PRGSTRC1
ALLCALLS)
```

The upper portion of the printout is the usual horizontal (teletype) version of the tree:



This tree is straightforwardly derived from the listing of the functions: PRINTSTRUCTURE calls only three functions PRINTSTRUC, PROGSTRUC and ALLCALLS in that order; PRINTSTRUC only calls itself,

and PROGSTRUC only PRGSTRC; PRGSTRC calls four functions, NOTFN, PROGSTRUC, PRGSTRC1 and PRGSTRC. Note that a function whose substructure has already been shown is not expanded in its second occurrence in the tree. The list following the tree printout represents a "reasonable" order for the listing of functions, and is available as the value of the variable DONE, used free in the program. The variable TREE contains a list structure version of the tree printed out.

The function NOTFN is a predicate which determines which functions should (not) be listed. It has value T for input an atom which should not be listed as a subfunction, either because this atom is a system function, or not defined as a function at all. In the BBN-LISP system, the function GETD (for getting function definitions) returns NIL for undefined functions, and a number for system and compiled functions. Therefore, for NOTFN we can use the simple definition (ATOM(GETD X)) as shown in the listing. An alternative definition would utilize a list of "interesting" functions; NOTFN would then return T if the input atom were not a member of this list. Only these "interesting" functions would then appear in the tree.

In addition to the tree, it is often important to know all functions which call a specified function. PRINTSTRUCTURE can be given a second (optional*) argument which should be a name, FN1, of a function which can be reached from FN. PRINTSTRUCTURE uses the list structure tree which is the value of TREE to determine the set of functions which call FN1 (which can be gotten to from FN). If the first argument to PRINTSTRUCTURE is NIL, it uses the value of the tree computed earlier. An example is:

```
←PRINTSTRUCTURE(NIL PRGSTRC)

PRGSTRC IS CALLED BY:
(PROGSTRUC PRGSTRC1 PRGSTRC)
```

*In BBN-LISP if a function is given fewer arguments than required, NIL values are provided.

LISTING

```
(PRINTSTRUCTURE
  (LAMBDA (FN FN1)
    (COND
      (FN (SETQ DONE NIL)
        (TERPRI)
        (PRINTSTRUC (SETQ TREE (CAR (PROGSTRUC FN))))
        (TERPRI)
        (SETQ DONE (REVERSE DONE))))
    (COND
      (FN1 (SETQ ALLCALLS NIL)
        (TERPRI)
        (PRIN1 FN1)
        (PRIN1 (QUOTE " IS CALLED BY:"))
        (TERPRI)
        (DREVERSE (ALLCALLS FN1 TREE)))
      (T DONE))))
```

(LB1, p7)

```
(PRINTSTRUC
  (LAMBDA (X N)
    (COND
      ((NULL N)
        (SETQ N 0)))
    (SPACES (DIFFERENCE N (POSITION)))
    (COND
      ((ATOM X)
        (PRINT X))
      ((NULL (CDR X))
        (PRINT (CAR X)))
      (T (PRIN1 (CAR X))
        (SPACES 2)
        (SETQ N (POSITION))
        (MAPC (CDR X)
          (FUNCTION (LAMBDA (Z)
              (PRINTSTRUC Z N)))))))))

(PROGSTRUC
  (LAMBDA (X D)
    (SETQ DONE (CONS X DONE))
    (LIST (CONS X (PRGSTRC (GETD X))))))

(PRGSTRC
  (LAMBDA (X A)
    (COND
      ((ATOM X)
        (COND
          ((NOTFN X)
            NIL)
          ((MEMB X D)
            NIL)
          ((MEMB X DONE)
            (SETQ D (CONS X D))
            (LIST X))
          (T (SETQ D (CONS X D))
            (PROGSTRUC X NIL))))
      (T (SELECTQ (SETQ A (CAR X))
          ((LAMBDA NLAMBDA
              PROG)
            (PRGSTRC1 (CDR X)))
          (COND
            (PRGSTRC1 X))
          (FUNCTION (PRGSTRC (CADR X)))
          (NCONC (PRGSTRC A)
            (PRGSTRC1 X)))))))
```

(LB1, p8)

```
(NOTFN
  (LAMBDA (FN)
    (ATOM (GETD FN))))

(PRGSTRC1
  (LAMBDA (L)
    (PROG (A B)
          (SETQ A (LIST NIL))
      LP  (COND
            ((NULL (SETQ L (CDR L)))
              (RETURN (CAR A)))
            ((NOT (ATOM (SETQ B (CAR L))))
              (LCONC (PRGSTRC B)
                A)))
          (GO LP)
      )))

(ALLCALLS
  (LAMBDA (FN TR)
    (PROG (A B)
          (SETQ A (CAR TR))
      LP  (COND
            ((NULL (SETQ TR (CDR TR)))
              (RETURN ALLCALLS))
            ((EQ (SETQ B (CAR TR))
                 FN)
              (SETQ ALLCALLS (CONS A ALLCALLS))
              (GO LP))
            ((ATOM B)
              (GO LP))
            ((EQ (CAR B)
                 FN)
              (SETQ ALLCALLS (CONS A ALLCALLS))))
          (ALLCALLS FN B)
          (GO LP)
      )))
```

Editor's Note

The following paper represents a first (good) attempt to introduce some standardization into LISP for the purpose of communication. I fully support this standard, but recommend that it be expanded to include some new features, including the (implicit) progn described earlier. Readers comments on this topic will be appreciated.

(LB1, p9)

STANDARD LISP

Hearn, Anthony C. (Alfred P. Sloan Foundation Fellow) Stanford Artificial Intelligence Report (Memo AI-90), May 1969. Research sponsored in part by the Air Force Office of Aerospace Research, U.S. Air Force, under AFOSR Contract No. F44620-68-C-0075, and in part by the Advance Research Projects Agency of the Secretary of Defense (SD-183).

SECTION I. INTRODUCTION

When it was first formulated in 1960, (1) the programming language LISP was a truly machine independent language. However, even the earliest computer implementation encountered problems in input-output control and the handling of free variables which were not considered in the original paper. Successive implementations of LISP on more sophisticated machines have solved such problems by independent methods and introduced extensions of the language peculiar to those machines. Consequently, a LISP user now faces considerable difficulty in moving a program from one machine to another and is often involved in weeks of debugging in the process. As a possible solution to this problem, this paper is an attempt to provide a uniform subset of LISP 1.5 and its variants as it exists today. The version of LISP described, which we call Standard LISP, is sufficiently restricted in form so that programs written in it can run under any LISP system upwardly compatible with LISP 1.5 as described in the LISP 1.5 Programmer's Manual (2). As function names vary from system to system and input-output control is different, some modification of the code is of course necessary before function definitions can be successfully compiled in any given system. However, this modification is performed automatically by a preprocessor, which is custom built for a particular system. This preprocessor is a LISP program which is loaded before any Standard LISP programs are run, and could be built automatically into a system if only Standard LISP programs are run. Parts of this preprocessor are similar for all systems, but some of it is peculiar to a given implementation. Standard LISP preprocessors have been written for SHARE LISP for the IBM 7090 series machines, Stanford LISP/360 for IBM System 360 machines, Stanford AI LISP 1.6 for the PDP-6 and PDP-10, BBN-LISP for the SDS 940 and Texas LISP for the CDC 6600. For convenience in exposition we shall refer to the first four systems as SHARE LISP, LISP/360, PDP LIST and BBN-LISP respectively.

In Section 2 of this paper, the structure of Standard LISP programs is described. Standard LISP conforms as closely as possible to LISP 1.5 as defined in the LSIP 1.5 Programmer's Manual, and the necessary deviations and extensions are described in detail. In Section 3, the structure of the Standard LISP Preprocessor is described, and the preprocessor for LISP/360 is given in Appendix A as an example. The translation of general LISP programs into Standard LISP is also discussed in Section 3. A listing of all functions defined in Standard LISP is given in Appendix B, and a function for reading Standard LISP programs is given finally in Appendix C.

The naming of new functions in a language and their definition is of course a very subjective matter, but little justification will be offered for their choice. The author would however appreciate hearing from anyone with criticisms or suggestions for improvements in the formulation.

SECTION 2. STRUCTURE OF STANDARD LISP PROGRAMS

2.1 Preliminary

In order to achieve the greatest possible compatibility with existing LISP systems, Standard LISP is based as closely as possible on the language described in the LISP 1.5 Programmer's Manual. However, there are six main areas where Standard LISP makes significant departures from the description in the Manual in order to offer maximum flexibility in programming. These areas are as follows:

(i) Handling of free variables and constants
(ii) Functional arguments
(iii) Character reading and printing
(iv) External file management
(v) Function and MACRO definitions
(vi) Array handling

Each of these modifications will be described in detail in subsequent parts of this Section. A number of additional limitations forced on Standard LISP programmers because of deficiencies in the design of various systems are also given in this Section.

Several functions given in the LISP 1.5 Programmer's Manual have been redefined in Standard LISP for maximum compatibility with other systems and several additional functions of proven utility have been included. These functions are defined in detail in Appendix B and are therefore not discussed in this Section.

2.2 Free Variables

One area in which the greatest differences between various LISP implementations can be seen is in the handling of free variables. Most systems allow the use of a special cell (usually under an APVAL or VALUE property) for storing those free variables which are global to all functions or are constant in the system. In systems with compilers there is in addition a mechanism for storing and retrieving free variables in compiled functions on a push-down stack. However, communication between a free variable in a compiled and an interpreted function is not possible in general in such a system, as the mechanism for handling each is different. Two systems (PDP LISP and BBN-LISP) solve this problem by making all interpreter variables SPECIAL and using a push-down-stack rather than an ALIST for storing their bindings. Standard LISP, however, cannot assume that such communication is possible and therefore imposes certain limitations in the use of free variables on the user. In order to provide a partial solution, however, three classes of free variables are recognized, each of which is handled differently by the preprocessing stage of an assembly. The three classes are:

(i) Constant variables or constants are variables which are global to all functions and whose value remains fixed during the life of a given system.

(ii) Global variables are variables which are global to all functions in that they do not appear in the variable list of any LAMBDA or PROG expression.

(iii) Extended variables are variables which are bound in a LAMBDA or PROG variable list of some function, but free in another.

Constant variables are declared by the function CONSTANT during the preprocessing stage. CONSTANT takes a list of pairs of constants and their values as its arguments, and is defined in Standard LISP as

constant[u] - deflist[u;CONSTANT]

The preprocessor replaces all such constants by the list

(QUOTE<value>)

on assembly. This mechanism provides a convenient method for compensating for system differences in the handling of atomic symbols containing arbitrary characters. For example, the atom 'AN ATOM' would be written as $$$AN ATOM$ in SHARE LISP and LISP/360, and as AN/ATOM in PDP LISP, so such a string cannot be introduced directly into a standard LISP program. However, the programmer can use a free variable ANATOM for example to represent this string, and the appropriate value for the atom declared during the setup stage. Any necessary character value objects, such as COMMA and LPAR, for example, should also be declared using CONSTANT. An example of the call of CONSTANT for LISP/360 would be:

CONSTANT((COMMA $$$,$) (LPAR $$$($) (ANATOM $$$AN ATOM$)))

Global variables differ from extended variables in that only one fixed cell is necessary to store the value pointer for the variable, and therefore no push-down-stack/ ALIST conflict arises. Thus it is possible to communicate between such variables in interpreted and compiled functions provided that references to such variables in interpreted code are changed to references to the particular cell where the value pointer is stored. In most systems, any such variables which appear in compiled functions must be declared SPECIAL before compilation. The SPECIAL declaration initializes such variables to NIL and sets up the necessary cell for storing the value pointer. Standard LISP introduces two functions GTS and PTS which allow the programmer to get or change the value of these special cells directly. In SHARE LISP for example these functions are defined as follows:

```
gts[u] = cdar[prop[u; SPECIAL ;λ[NIL;error[cons[u;(NOT SPECIAL)]]]]]
pts[u;v] = rplacd[car[prop[u;SPECIAL;λ[NIL;list[put[u;SPECIAL;
                                                list[NIL]]]]]];v]
```

in other systems, equivalent definitions are always possible. If a given function definition is to be interpreted and not compiled, then the preprocessor changes every reference to global variables in the definition to calls to GTS and PTS. No action is necessary by the preprocessor if the function is compiled.

Extended variables are the only class of free variables which impose limitations on the user. In all systems known to the author such variables require SPECIAL declaration before compilation of functions using them. However, it is not in general possible to mix interpreted and compiled functions using these variables, and so functions containing such variables should either be all compiled or all interpreted.

All references to global and extended variables are made directly to the variable in Standard LISP unless the user wishes to make an explicit call using GTS. Similarly all changes should be made using SETQ, or PTS for explicit references. The functions CSET, CSETQ and SET are consequently not defined or available in Standard LISP.

2.3 The Free Variables ALIST and OBLIST

Many systems allow the user to reference the ALIST and OBLIST directly. However, the structure and referencing (and even existence) of these lists vary so much from system to system that no direct reference can be made in Standard LISP. The interpreter functions EVAL and APPLY are however still available as *EVAL and *APPLY. *EVAL takes a single form as argument and returns its evaluated value. *APPLY takes two arguments, a function and a list of arguments for that function, respectively and returns the value of applying the function to the argument list. Thus in SHARE LISP these two functions are defined as:

```
*eval[u] = eval[u;alist[]]
*apply[u;v] = apply[u;v;alist[]]
```

where alist [] is defined by

```
LAP ((ALIST SUBR Ø)(CLA $ALIST)(TRA 1 4))NIL)
```

2.4 Functional Arguments

Incorporating functional arguments in LISP systems poses many design problems which are still not completely resolved. However, most systems recognize the difference between a call to a functional argument which is essentially a quoted function definition, and one in which the form of the functional argument changes during the evaluation, as given in Saunders' famous example (3) for instance. Standard LISP uses the PDP LISP technique for distinguishing between these two types of calls, viz., a quoted definition as referred to by FUNCTION and a modifiable form by *FUNCTION. The uses of QUOTE to define functional arguments is not allowed. The preprocessor can of course modify these forms to whatever particular calling method is used in the relevant system. Free variables in functional arguments which are not constants require SPECIAL or COMMON declarations in most compiler systems. These declarations are made in a user-defined initialization stage of the preprocessor as described in Section 3.

2.5 Character Reading and Printing

In its purest formulation, the constituent characters of a LISP atom have no inherent meaning. However many applications of LISP require the inspection and manipulation of these characters and the ability to create new atoms from a list of characters. It is also often necessary to read and write individual characters and exercise some control over the format of output.

Standard LISP introduces the following functions for this purpose. All can be readily defined in most systems.

readch[]	Reads and returns one character from the input buffer.
princ[u]	Adds the character string u to the output buffer. Returns u.
explode[u]	Returns a list of the constituent character atoms comprising the literal atom u. u may be an integer, but not a floating point number, as various systems use different print representations for these.
compress[u]	Creates an atom (literal atom or number) from the list of characters u, adds it to the OBLIST if one exists, and returns the atom.
liter[u]	A predicate function which is true if u is one of the character atoms A through Z and false otherwise.
digit[u]	A predicate function which is true if u is one of the character atoms 0 through 9.
otll[u]	If u = NIL then this function returns the current length of the output buffer line, otherwise it sets the buffer length to u.
pos[]	Returns the number of characters presently in the output buffer.
spaces[n]	Adds n spaces to the output buffer and returns NIL.

In order to achieve compatibility with all systems it is necessary to make a special distinction between character atoms as such and ordinary atoms. A character atom is one returned by READCH or in the list returned by EXPLODE. The functions LITER, and DIGIT should only be used with character atoms as arguments and COMPRESS can only take a list of character atoms. Character atoms do not possess property lists in the usual sense and bear no relation to the ordinary atom of the same name. Thus the character atom A is not equal in any sense to the atom A or the character atom 1 to the LISP number 1. If particular character atoms are needed in a program, they should be included as constants (e.g., ONE) in the main program and appropriately defined in the preprocessor. Alternatively, they can be created using EXPLODE.

2.6 File Management

The ability of most modern operating systems to handle varying input and output devices for data was not foreseen in the original definition of LISP. Standard LISP therefore includes the following functions to handle such file management. It is assumed that in each system there is a standard input and output device initially offered to the user.

open[u;INPUT]	Declares the file u as an input file.
open[u;OUTPUT]	Declares the file u as an output file.
rds[u]	Declares that all input now comes from the file u. If u is NIL, then the standard input device is selected. If u is not NIL, then u must have previously been declared through an OPEN statement as an input file.

wrs[u] Declares that all output now goes to the file u. If u is NIL, then the standard output device is selected. If u is not NIL, then u must have been previously declared through an OPEN statement as an output file.

close[u] Closes the file u. If u is an output file, then the necessary end-of-file marks are written.

The form of the argument u for these functions will vary from system to system. In some, it will be an atom specifying the name of a file in some unique manner. In others it will be a list giving details of device, filename, project/programmer numbers, etc. However, the differences should not affect the design of programs or their calling sequences.

2.7 User-defined Functions and Macros

Most LISP systems provide the user with a function DEFINE for introducing function definitions into the system. However, the action of DEFINE varies markedly from system to system. In SHARE LISP-like systems, the definition is added to the property list of the function name with an indicator EXPR. Other systems use CAR of the function name as a pointer to the definition and in some cases compile the function directly into the system. A Standard LISP function DEFINE is also available for introducing user-defined functions into the system, but as its action varies from system to system its definition must be included in the system preprocessor. The Standard LISP DEFINE normally includes a call to the system compiler, but this may be varied as required. In addition, DEFINE calls the preprocessor translator which modifies the function definition before incorporation in the system as we shall explain in detail in Section 3.

Special forms introduced as FEXPRs and FSUBRs may also be defined in Standard LISP. As the preprocessor also modifies the definitions of these forms, a special function DEFEXPR, whose argument has the same form as DEFINE, is provided for introducing them.

The advantages of FEXPRs and FSUBRs are not universally recognized among LISP system designers, and the current tendency is to incorporate a macro defining facility in order to reduce execution time in compiled code. In line with this trend, Standard LISP also includes a function MACRO similar in form to that described by Weissman (4). As with DEFINE and DEFEXPR, the argument list is modified by the preprocessor and the code is compiled if required. In systems with no macro handing facility the preprocessor can be used to expand any macros found in a function definition at DEFINE time as shown in Appendix A. On the other hand, if the system has no FEXPR or FSUBR facility, such forms can be handled with macros or again by the preprocessor. Consequently, all macros and special forms must be in the system before defining any function which uses them.

Because function definitions are stored differently from system to system, Standard LISP includes a function GETD for retrieving a

function definition when required. With interpreted code, GETD returns a pointer to the S-expression definition of the function. If the function has been compiled, GETD still returns a non-NIL value so that a test for an existing function definition can be made, but it is not possible to interpret this value in Standard LISP.

2.8 Array Handling

Many LISP systems recently developed allow for the definition of arrays in which the elements may be numbers as well as pointers to S-expressions. In order to maintain our downwards compatibility with SHARE LISP, however, Standard LISP restricts arrays to the LIST array described in the LISP 1.5 Programmer's Manual. Consequently the word LIST in the SHARE LISP array declaration is redundant, and is omitted in Standard LISP. Thus ARRAY is a function of one argument which is a list of arrays to be declared. Each item is a list containing the name of the array and its dimensions, an example being

```
array [((ALPHA (7 10)) ((BETA (3 4 5))]
```

After ARRAY has been executed, the arrays declared exist and their elements are all set to NIL. Indices always range from 0 to n-1.

On the other hand, Standard LISP provides two distinct functions for setting array elements and getting their values. SETEL is a function of two arguments, the first a list consisting of the array name and the relevant coordinates, and the second the value of the element. Similarly, GETEL is a function of one argument which returns the value of an element. For example, to set the (3,4) element of the array ALPHA to A, we write

```
setel [(ALPHA 3 4) ; A]
```

and to get the value of the (0,1,2) element of BETA we write

```
getel [(BETA 0 1 2)]
```

2.9 Standard LISP Program Restrictions

In order to achieve compatibility with as many systems as possible, the following additional restrictions and features must be borne in mind when writing programs in Standard LISP.

(i) The programmer is talking to EVALQUOTE rather than EVAL. Moreover, there are no OVERLOAD cards in the SHARE LISP sense, so that a special reading function may be required in some systems. A suitable function to do this is given in Appendix C.

(ii) Literal atoms are limited to 24 characters. Decimal integers are restricted to 7 digits ($2^{\pm 23}$) floating point numbers to 8 digits plus an optional signed or unsigned exponent less in magnitude than 37. Numbers relating to other bases are not allowed explicitly and must be included as constants.

(iii) Functions may not have more than five arguments and be compiled in PDP LISP.

(iv) A GO statement may only occur at the top level of a PROG or as one of the value parts in a top level COND statement within a PROG (otherwise the function will not compile in the 7090 and LISP/360 systems).

(LBl, pl6)

(v) The atom NIL represents falsity; F is not available for this purpose. Furthermore, all S-expressions other than NIL are considered to be true in predicate tests.

(vi) A COND form must terminate with a pair (T <form>) unless the COND occurs within a PROG form.

(vii) Free variables should not have the same name as a function in the program (otherwise some compilers will fail).

(viii) No atom may have more than 24 characters.

(ix) Because the Standard LISP function DEFINE includes a call to the preprocessor, all FEXPRS and MACROS must be in the system before defing any function which uses them.

(x) LABEL is not defined in Standard LISP.

SECTION 3. THE STANDARD LISP PREPROCESSOR

3.1 Preliminary

The Standard LISP preprocessor is a LISP program which is responsible for modifying any functions or MACROS defined in Standard LISP so that they conform with the particular properties of the system in which they are being assembled. The preprocessor is written directly in the particular LISP language under consideration and we cannot therefore describe in detail its form for each system. However, the general characteristics remain the same and we shall describe them here.

The preprocessor divides naturally into three parts as follows:

(i) The translator, which modifies the S-expression definitions of functions before they are compiled or interpreted.

(ii) Definition of functions in Standard LISP not implemented in the particular system.

(iii) A user supplied initialization section where free variables are declared.

3.2 The Translator

The Standard LISP translator is a recursive function TRANS which modifies the code of any function or macro introduced into the system. TRANS takes two arguments, the first being the expression to be modified and the second NIL if the function is to be compiled and T if not. The code for the LISP/360 translator is given in Appendix A. This code is intended only as a guide for writing preprocessors for other systems, but in most cases it will not require revision except as indicated below.

In order to use the function TRANS, it is necessary to define in the preprocessor the functions DEFINE, DEFEXPR and MACRO. These functions call an auxiliary function DEFl which performs four separate actions on each definition considered.

(i) It checks each function name for a flag LOSE. If the flag is found, it does not introduce the function into the system. In this way any functions not needed in a particular implementation of a large system may be excluded without modifying the main program.

(ii) If the function has no such flag, DEFl now checks for the existence of a functional property indicator. If it finds one, a message

(***** <function name> REDEFINED)

is printed. This is very useful in checking for conflicts between LISP system functions and Standard LISP functions. If a conflict is found, the relevant function can be renamed using the NEWNAM mechanism described below.

(iii) It now applies the function TRANS to the function name and its definition. The explicit action of TRANS is described below.

(iv) The function is now compiled into the system, using in most systems the equivalent of the function COM1 of the SHARE LISP compiler, or interpreted if the function has the flag NOCOMP on its property list or a global variable NOCMP is true. This code may need modification in some systems. In LISP/360, COM1 takes three arguments, the first the function name, the second the definition of the function if it is an EXPR (or NIL) and the third the definition if it is an FEXPR (or NIL).

The necessary flags for functions which are not to be included in an assembly or are not to be compiled are added by the functions LOSE and NOCOMP respectively. Each of these functions takes a list of function names as argument. Their definitions are given in Appendix A, and their use is obvious.

As an alternative to flagging those functions which are not to be compiled, it may sometimes prove useful to interpret a whole batch of functions. In this case, the global variable NOCMP can be turned on and off by the function NOCOM also defined in Appendix A.

After all functions have been considered in a given DEFINE or MACRO statement, a list of those functions which have been compiled into the system is returned (excluding those 'lost').

3.3 Modifications to Code Performed by TRANS

Besides replacing any constants declared by CONSTANT with the corresponding quoted values as described in Section 2.2, and expanding any FEXPRs or macros in systems without facilities for handling them, TRANS performs two other types of code and modification on any definition given to it. These modifications are declared in advance by the functions NEWNAM and NEWFORM.

NEWNAM, like CONSTANT, takes a list of pairs of atoms which it gives to DEFLIST with the indicator NEWNAM. If TRANS meets any atoms with this property in a functional position, it replaces them by the corresponding value. This mechanism is useful for two purposes:

(i) To rename those functions in a Standard LISP program whose names conflict with LISP system functions. In some cases, however, it will be necessary to define the renamed Standard LISP function in terms of the function with the old name in the LISP system. This can be done using the NEWFORM mechanism described below, or by renaming the functions with the NEWNAM mechanism and then defining them without applying the translator to their definition. An example of this is the function EXPLODE in the LISP/360 Preprocessor in Appendix A.

(ii) To rename those functions in a Standard LISP program whose definitions coincide with system functions of a different name. Examples of this may be found in the call of NEWNAM in the LISP/360 Preprocessor.

NEWFORM takes a list of pairs of atoms and lambda functional definitions which it gives to DEFLIST with the indicator NEWFORM. When TRANS meets an atom in a functional position with such a property it modifies its translated arguments according to the prescription given by the lambda expression. This mechanism again has two uses:

(i) To redefine Standard LISP functions in terms of LISP system functions whose definitions (and maybe even name) coincide, but the arguments are in a different order. For example, PDP LISP contains a function PUTPROP which is the same as the Standard LISP function PUT except that the second and third arguments are in the opposite order. So in assembling Standard LISP programs in this system, NEWFORM has an entry

(PUT (LAMBDA (U V W) (PUTPROP U W V)))

We note, however, that this translation will not be correct if a form depends upon the order of its arguments for its effect. For example, the Standard LISP form

(PUT U (SETQ V (QUOTE NO)) V)

would not translate correctly into PDP LISP using the NEWFORM entry above.

(ii) To 'open compile' functions with a short definition in the particular LISP implementation. Examples of this are shown in the LISP/360 preprocessor.

If the second argument of TRANS is true, indicating that the function is not to be compiled, then the translator checks for occurrences of special variables appearing alone or as the first argument of SETQ. In the former cases, it replaces the code by a call to PTS and in the latter a call to GTS. In some systems, however, it may be easier to declare such variables as COMMON and use CSETQ to set their values. A check for common variables is therefore included in the definition of TRANS for illustrative purposes.

3.4 Translator Associated Functions

In addition to the functions described above, references are made to the functions SUBLIS, PAIR and DEFLIST defined in Standard LISP. If these functions are not present in a given system, their definitions must be included with the translator, as with SUBLIS in the LISP/360 Translator in Appendix A.

3.5 Standard LISP Functions Not in System

No LISP system yet written contains all Standard LISP functions exactly as we have defined them, and it will therefore be necessary to include such definitions in the preprocessor, either as a function definition compiled directly into the system without using TRANS, or by using the NEWNAM and NEWFORM mechanisms described earlier. Examples of this are shown in Appendix A.

3.6 User-supplied Preprocessor Section

The final section of the preprocessor initializes a user's free variables and must be supplied by the user for a given program. This section will contain the table of constants which the programmer uses as a call to CONSTANT, and also any necessary declaration of

other free variables which the system requires. These latter declarations cannot form part of the Standard LISP program, as the declaration mechanism varies from system to system. However, Standard LISP does allow use of a function SPECIAL to declare any free variables in function definitions generated by a user's program so that these functions may be compiled at the time of definition if required.

3.7 Program Translator

Because Standard LISP provides only a subset of the functions available in any given LISP system, it is not possible to automate completely the translation of any given LISP program into Standard LISP. However, it is usually only the system functions which deal with machine dependent properties which cannot be translated so that most programs can be converted fairly readily.

A program translator from PDP LISP to Standard LISP for example has been designed on similar lines to the Standard LISP preprocessor described in this Section. Apart from using reversed entries in the NEWNAM and NEWFORM tables, the program translator checks for three particular features in the PDP-LISP program being converted.

(i) Any quoted expressions are checked for characters which are neither letters or digits. If any are found, the quoted expression is replaced by a generated symbol.

(ii) Any forms containing functions with extended properties are converted by explicit routines designed for these functions. For example, a COND form can have more than one consequent in each term in PDP LISP, so that each COND expression must be checked and changed when necessary.

(iii) If a function is found which is not translatable, the form in which it occurs is replaced by a generated symbol.

Whenever replacement by a generated symbol occurs, the programmer is informed of this on his console. In addition, the generated symbol and its equivalent are placed on a table which is printed at the end of the translation. This table is searched for the prior existence of a non-translatable form before a new symbol is generated. In case (i) above, the generated symbol and its equivalent become CONSTANT entries in the Standard LISP preprocessor. In case (iii), the symbols can be replaced (using an editing program) by whatever Standard LISP form the programmer decides upon.

Program translators will be constructed for other systems as the need arises.

CONCLUSION

The formulation of LISP presented in this paper offers the user a language which may be run on most machines with a minimum of fuss. It has been successfully used to run one very large LISP program[5] in five different LISP systems without difficulty. We hope that this attempt at standardization will also have some influence on the design of new LISP systems and make easier the description of LISP algorithms and programs in the literature.

REFERENCES

1. John McCarthy, Comm of the ACM, 3, 184 (1960).

2. John McCarthy, Paul W. Abrahams, Daniel J. Edwards, Timothy P. Hart, Michael I. Levin, LISP 1.5 Programmer's Manual, MIT Press, 1965.

3. Robert Saunders, LISP - On the Programming System, in The Programming Language LISP: Its Operation and Applications, edited by E.C. Berkeley and D.G. Bobrow, MIT Press, (1964).

4. Clark Weissman, LISP 1.5 Primer, Dickenson, 1967.

5. Anthony C. Hearn, REDUCE - A User Oriented Interactive System for Algebraic Simplification, in Interactive Systems for Experimental Applied Mathematics, (Academic Press, New York, 1968).

6. Daniel G. Bobrow, Daniel L. Murphy, and Warren Teitelman, BBN-LISP, Bolt Beranek and Newman Inc., Cambridge, Massachusetts, 1969.

(LB1, p21)

APPENDIX A

*** THE STANDARD LISP PREPROCESSOR ***
*** FOR LISP/360 ***

```
DEFLIST (((COMMENT (LAMBDA (U A) NIL))) FEXPR)

COMMENT (STANDARD LISP TRANSLATOR)

SPECIAL ((NOCMP))

(LAMBDA (U) (COMPILE (DEFINE U))) ((

(DEFINE (LAMBDA (U)
    (DEF1 U (QUOTE EXPR))))

(DEFEXPR (LAMBDA (U)
    (DEF1 U (QUOTE FEXPR))))

(MACRO (LAMBDA (U)
    (DEF1 U (QUOTE MACRO))))

(DEF1 (LAMBDA (U V)
    (PROG (W X Y Z)
    A  (COND ((NULL U) (RETURN Y)))
       (SETQ X (CAAR U))
       (COND ((FLAGP X (QUOTE LOSE)) (GO B))
             ((GETD X)
                (PRINT (LIST (QUOTE *****) X (QUOTE REDEFINED)))))
       (SETQ Y (NCONC Y (LIST X)))
       (SETQ W (OR NOCMP (FLAGP X (QUOTE NOCOMP))))
       (SETQ X (LIST (TRANS X NIL) (TRANS (CADAR U) NOCOMP)))
       (COND ((OR W (EQ V (QUOTE MACRO))) (DEFLIST (LIST X) V))
             (T (COM1 (CAR X)
                      (AND (EQ V (QUOTE EXPR)) (CADR X))
                      (AND (EQ V (QUOTE FEXPR)) (CADR X)))))
    B  (SETQ U (CDR U))
       (GO A))))

(TRANS (LAMBDA (U V)
    (COND ((NULL U) NIL)
          ((ATOM U) (COND ((NUMBERP U) U)
                          ((AND V (GET U (QUOTE SPECIAL)))
                             (LIST (QUOTE GTS) (LIST (QUOTE QUOTE) U)))
                          (T ((LAMBDA (X)
                             (COND (X (LIST (QUOTE QUOTE) X))
                                   (T ((LAMBDA (Y)
                                        (COND (Y Y) (T U)))
                                       (GET U (QUOTE NEWNAM))))))
                             (GET U (QUOTE CONSTANT))))))
          ((ATOM (CAR U))
             (COND ((EQ (CAR U) (QUOTE QUOTE)) U)
                   ((AND V (EQ (CAR U) (QUOTE SETQ)))
                      (APPEND (COND ((GET (CADR U) (QUOTE SPECIAL))
                                     (LIST (QUOTE PTS)
                                       (LIST (QUOTE QUOTE) (CADR U))))
```

(LB1, p22)

```
                                        ((FLAGP (CADR U) (QUOTE COMMON))
                                         (LIST (QUOTE CSETQ) (CADR U)))
                                        (T (LIST (QUOTE SETQ) (CADR U))))
                                    (TRANS (CADDR U) T)))
                        (T (PROG (X)
                             (RETURN (COND
                                ((SETQ X (GET (CAR U) (QUOTE NEWFORM)))
                                 (SUBLIS (PAIR (CADR X) (MAPTR (CDR U) V))
                                         (CADDR X)))
                                ((SETQ X (GET (CAR U) (QUOTE NEWNAM)))
                                 (CONS X (MAPTR (CDR U) V)))
                                ((SETQ X (GET (CAR U) (QUOTE MACRO)))
                                 (TRANS (APPLY X (CDR U) NIL)))
                                (T (CONS (CAR U) (MAPTR (CDR U) V)))))))))
                (T (MAPTR U V)))))

(MAPTR (LAMBDA (U V)
    (COND ((ATOM U) (TRANS U V))
          (T (CONS (TRANS (CAR U) V) (MAPTR (CDR U) V))))))

(CONSTANT (LAMBDA (U)
    (DEFLIST U (QUOTE CONSTANT))))

(LOSE (LAMBDA (U)
    (FLAG U (QUOTE LOSE))))

(NEWFORM (LAMBDA (U)
    (DEFLIST U (QUOTE NEWFORM))))

(NEWNAM (LAMBDA (U)
    (DEFLIST U (QUOTE NEWNAM))))

(NOCOMP (LAMBDA (U)
    (FLAG U (QUOTE NOCOMP))))

(NOCOM (LAMBDA (U)
    (SETQ NOCMP U)))
))

UNSPECIAL ((NOCMP))

(LAMBDA (U) (COMPILE (DEFLIST U (QUOTE EXPR)))) ((

(GETD (LAMBDA (U)
    (OR (GET U (QUOTE EXPR))
        (GET U (QUOTE FEXPR))
        (GET U (QUOTE SUBR))
        (GET U (QUOTE FSUBR))
        (GET U (QUOTE MACRO)))))

(SUBLIS (LAMBDA (U V)
    (COND ((NULL U) V)
          (T ((LAMBDA (X)
               (COND (X (CDR X))
                     ((ATOM V) V)
                     (T (CONS (SUBLIS U (CAR V)) (SUBLIS U (CDR V))))))
              (SASSOC V U NIL))))))
))
```

(LB1, p23)

```
CONSTANT ((
     (BLANK $$$ $)
     (COMMA $$$,$)
     (DOLLAR $$/$/)
     (LPAR $$$($)
     (RPAR $$$)$)
     (STAR $$$*$)
))

NEWNAM ((
     (DIGIT DIGP)
     (EXPLODE *EXPLODE)
     (LITER LETP)
     (OPEN *OPEN)
     (OTLL *OTLL)
     (PRINC PRIN1)
     (RDS *RDS)
     (SPACES XTAB)
     (WRS *WRS)
     (*FUNCTION FUNCTION)
))

NEWFORM ((
     (*APPLY (LAMBDA (U V) (APPLY U V ALIST)))
     (CAAAAR (LAMBDA (U) (CAAR (CAAR U))))
     (CAAADR (LAMBDA (U) (CAAR (CADR U))))
     (CAADAR (LAMBDA (U) (CAAR (CDAR U))))
     (CAADDR (LAMBDA (U) (CAAR (CDDR U))))
     (CADAAR (LAMBDA (U) (CADR (CAAR U))))
     (CADADR (LAMBDA (U) (CADR (CADR U))))
     (CADDAR (LAMBDA (U) (CADR (CDAR U))))
     (CADDDR (LAMBDA (U) (CADR (CDDR U))))
     (CDAAAR (LAMBDA (U) (CDAR (CAAR U))))
     (CDAADR (LAMBDA (U) (CDAR (CADR U))))
     (CDADAR (LAMBDA (U) (CDAR (CDAR U))))
     (CDADDR (LAMBDA (U) (CDAR (CDDR U))))
     (CDDAAR (LAMBDA (U) (CDDR (CAAR U))))
     (CDDADR (LAMBDA (U) (CDDR (CADR U))))
     (CDDDAR (LAMBDA (U) (CDDR (CDAR U))))
     (CDDDDR (LAMBDA (U) (CDDR (CDDR U))))
     (DIVIDE (LAMBDA (U V) (CONS (QUOTIENT U V) (REMAINDER U V))))
     (ERRORSET (LAMBDA (U V) (LIST U)))
     (PUT (LAMBDA (U V W) (PROG2 (DEFLIST (LIST (LIST U W)) V) W)))
     (GENSYM (LAMBDA NIL (GENSYM1 (QUOTE $$$   G$))))
     (ONEP (LAMBDA (N) (EQUAL N 1)))
     (READCH (LAMBDA NIL (READCH NIL)))
))

COMMENT (STANDARD LISP FUNCTIONS DEFINED IN TERMS OF SYSTEM
         FUNCTIONS OF THE SAME NAME)

COMMENT (THE FOLLOWING LIST IS USED BY EXPLODN1 DEFINED BELOW)

DEFLIST (((NASL (((0 . $$$0$) (1 . $$$1$) (2 . $$$2$) (3 . $$$3$)
                  (4 . $$$4$) (5 . $$$5$) (6 . $$$6$) (7 . $$$7$)
                  (8 . $$$8$) (9 . $$$9$))))) SPECIAL)
```

(LB1, p24)

```
(LAMBDA (U) (COMPILE (DEFLIST U (QUOTE EXPR)))) ((

(*EXPLODE (LAMBDA (U)
    (COND ((NUMBERP U) (EXPLODN U))
          (T (EXPLODE U)))))

(EXPLODN (LAMBDA (U)
    (COND ((ZEROP U) (LIST (QUOTE $$$0$)))
          ((NOT (FIXP U)) (ERROR (LIST (QUOTE EXPLODN) U)))
          ((MINUSP U) (CONS (QUOTE $$$-$) (EXPLODN (MINUS U))))
          (T (EXPLODN1 U)))))

(EXPLODN1 (LAMBDA (U)
    (PROG (X Y Z)
    A  (COND ((ZEROP U) (RETURN Z)))
       (SETQ X (REMAINDER U 10))
       (SETQ Y NASL)
    B  (COND ((EQUAL X (CAAR Y)) (GO C)))
       (SETQ Y (CDR Y))
       (GO B)
    C  (SETQ Z (CONS (CDAR Y) Z))
       (SETQ U (QUOTIENT U 10))
       (GO A))))

(*OPEN (LAMBDA (U V)
    (PROG2 (OPEN U (QUOTE ((LRECL . 80) (BLKSIZE . 80))) V)
           U)))

(*RDS (LAMBDA (U)
    (COND ((NULL U) (RDS (QUOTE LISPIN)))
          (T (RDS U)))))

(*WRS (LAMBDA (U)
    (COND ((NULL U) (WRS (QUOTE LISPOUT)))
          (T (PROG NIL (OTLL 72) (ASA NIL) (WRS U))))))
))

UNSPECIAL ((NASL))

COMMENT (STANDARD LISP FUNCTIONS NOT DEFINED IN LISP 360)

DEFINE ((

(COPY (LAMBDA (U)
    (COND ((ATOM U) U)
          (T (CONS (COPY (CAR U)) (COPY (CDR U)))))))

(COMPRESS (LAMBDA (U)
    (PROG2 (COND ((DIGIT (CAR U)) (MAPCAR U (FUNCTION RNUMB)))
                 (T (MAPCAR U (FUNCTION RLIT))))
           (MKATOM))))

(GTS (LAMBDA (U)
    ((LAMBDA (X)
       (COND ((NULL X) (ERROR (LIST (QUOTE GTS) U)))
             (T (CAR X))))
    (GET U (QUOTE SPECIAL)))))
```

(LB1, p25)

```
(PTS (LAMBDA (U V)
    (CAR ((LAMBDA (X) (COND ((NULL X) (PUT U (QUOTE SPECIAL) (LIST V)))
                            (T (RPLACA X V))))
          (GET U (QUOTE SPECIAL))))))

(MAP (LAMBDA (U PI)
    (PROG NIL
    A   (COND ((NULL U) (RETURN NIL)))
        (PI U)
        (SETQ U (CDR U))
        (GO A))))

(MAPCON (LAMBDA (U PI)
    (COND ((NULL U) NIL)
          (T (NCONC (PI U) (MAPCON (CDR U) PI))))))

(REVERSE (LAMBDA (U)
    (PROG (V)
    A   (COND ((NULL U) (RETURN V)))
        (SETQ V (CONS (CAR U) V))
        (SETQ U (CDR U))
        (GO A))))

(SUBST (LAMBDA (U V W)
    (COND ((EQUAL V W) U)
          ((ATOM W) W)
          (T (CONS (SUBST U V (CAR W)) (SUBST U V (CDR W)))))))

(*EVAL (LAMBDA (U)
    (EVAL U ALIST)))
))

DEFEXPR ((

(PROGN (LAMBDA (U A)
    (PROG NIL
    A   (COND ((NULL (CDR U)) (RETURN (EVAL (CAR U) NIL))))
        (EVAL (CAR U) NIL)
        (SETQ U (CDR U))
        (GO A))))
))


COMMENT (THE FUNCTIONS POS AND *OTLL AND THE ARRAY FUNCTIONS ARE
         DEFINED IN LAP AND NOT INCLUDED IN THIS LISTING)


COMMENT (E N D   O F   L I S P   3 6 0   P R E P R O C E S S O R)
```

(LB1, p26)

APPENDIX B

THE FOLLOWING FUNCTIONS DEFINED IN THE LISP 1.5 PROGRAMMER's MANUAL ARE AVAILABLE IN STANDARD LISP

ADD1	LENGTH	PRIN1
AND	SESSP	PRINT
APPEND	LIST	PROG
ATOM	LITER	PROG2
CAR...CDDDDR	LOGAND	QUOTE
COND	LOGOR	QUOTIENT
CONS	LOGXOR	READ
COPY	MAP	RECIP
DEFLIST	MAPCON	REMAINDER
DIFFERENCE	MAPLIST	REMFLAG
DIGIT	MAX	REMPROP
EQ	MEMBER	RETURN
EQUAL	MIN	REVERSE
ERROR	MINUS	RPLACA
EXPT	MINUSP	RPLACD
FIXP	NCONC	SASSOC
FLAG	NOT	SUB1
FLOATP	NULL	SUBLIS
GET	NUMBERP	SUBST
GO	ONEP	TERPRI
GREATERP	OR	TIMES
LEFTSHIFT	PAIR	ZEROP
	PLUS	

The following functions defined in Standard LISP have the same names but different properties from the LISP 1.5 Programmer's Manual definitions. These changes are necessary in order to provide for maximum compatibility with other system definitions.

array [u]	initializes arrays. Same definition as for SHARE LISP except that 'LIST' is omitted.
define [u]	introduces function definitions into system. Returns a list of function names introduced.
divide [u;v]	returns cons [quotient [u;v] ; remainder [u;v]] where u and v are integers.
errorset [u;v]	if an error occurs during the evaluation of u, NIL is returned. The error message is printed if and only if v is T. If no error occurs, list [u] is returned. In systems which lack this facility, errorset [u;v] may be replaced by list [u], but no error recovery is then possible except at the top level.

function [u]	defines a quoted functional argument
gensym []	creates a new atom (e.g., GØ123), adds it to the OBLIST if one exists and returns atom.
setq [u;v]	sets the PROG or free variable u to the value of v. Returns value of v.
special [u]	initializes the list of u of free variables for compilation. Returns u.

The following additional functions are defined in Standard LISP:

close [filename]	closes filename by writing appropriate end-of-file marks, etc. Returns filename.
compress [u]	creates an atom (literal atom or number) from list of characters u, adds it to the OBLIST if one exists, and returns the atom.
defexpr [u]	introduces special form definitions (as FEXPRs or FSUBRs) into system. Returns a list of function names introduced.
delete [u;v]	deletes the first top level occurrence of the S-expression u from the list v.
explode [u]	returns a list of the constituent characters of atom u. Must also work for integers. e.g. explode [123] = (1 2 3).
fix [u]	returns integer part of number u.
flagp [u;v]	returns T if atom u has flag v on its property list, otherwise NIL.
float [u]	Returns floating point equivalent of number u.
getd [u]	returns pointer to definition of u, if u is a function or macro, and NIL otherwise.
getel [[u;m_1,..,m_n]]	returns (m_1,...,m_n) element of array named u.
gts [u]	returns ('gets') the value of the special atom u.
macro [u]	introduces macro definitions into system. Returns a list of macro names introduced.
mapcar [u;fn]	if null [u] then NIL else cons [fn[car[u]]; mapcar [cdr [u]; fn]].
open [filename; stat]	opens a file named filename on some external device. stat = INPUT or OUTPUT. Returns filename.
princ [u]	adds character string u to output buffer. Returns u.

(LB1, p28)

pos []	returns number of characters presently in output buffer. Initialized to zero at every call of TERPRI.
progn[u_1,...,u_n]	evaluates forms u_1,...,u_n and returns value of u_n.
put [u; ind; prop]	puts prop on property list of u with indicator ind. Returns prop.
rds [filename]	selects filename as input device. All input now comes from filename. Returns filename.
setel [[u;m_1,...,m_n];v]	sets (m_1,...,m_n) element of array named u to v. Returns v.
spaces [n]	adds n spaces to output buffer. Returns n.
wrs [filename]	selects filename as output device. All output now goes to filename. Returns filename.
*apply [u;v]	applies function u to list of arguments v, and returns result.
*eval [u]	evaluates expression u (using current ALIST if one exists) and returns value.
*function [u]	defines a functional argument whose form may change during evaluation.

(LB1, p29)

APPENDIX C

The following function may be used to assemble a Standard LISP program in systems with OVERLOAD directives or EVAL listen loops. It is assumed that the atom STOP occurs at the end of the program.

```
(SREAD (LAMBDA NIL
 (PROG (X Y)
  A     (TERPRI)
        (COND
            ((NULL (SETQ X (ERRORSET (READ) T))) (GO ERR))
            ((EQ (SETQ X (CAR X)) (QUOTE STOP)) (BO B))
            ((OR (NULL (SETQ Y (ERRORSET (READ) T)))
                 (EQ (SETQ Y (CAR Y)) (QUOTE STOP))) (GO ERR)))
        (PRINC (QUOTE $$$FUNCTION...$))

        (PRINT X)
        (TERPRI)
        (TERPRI)
        (PRINC (QUOTE $$$ARGUMENTS... $))
        (PRINT Y)
        (TERPRI)
        (TERPRI)
        (SETQ Y (ERRORSET (*APPLY X Y) T))
        (COND ((NULL Y) (GO A)))
        (TERPRI)
        (PRINC (QUOTE $$$VALUE... $))
        (PRINT (CAR Y))
        (PRINT (CAR Y))
        (GO A)
  ERR   (TERPRI)
        (PRINC (QUOTE $$$READ ERROR$))
   B    (RETURN (QUOTE $$$*$)))))
```

(LB1, p30)

BALM - AN EXTENDABLE LIST-PROCESSING LANGUAGE

Harrison, Malcolm C. AEC Research and Development Report (NYO-1480-118). Courant Institute of Mathematical Sciences (New York University). 1969 June.

This paper describes an extendable list-processing system currently implemented on the CDC 6600 which includes the facilities provided by LISP 1.5, and permits the programmer to write in an Algol-like language. Additional data-types include vectors, strings, and entry-points. The language can be extended by adding new operators, and by specifying how expressions involving them should be translated into an intermediate language. (Abstract)

1. Introduction

The LISP 1.5 programming language[1] has emerged as one of the preferred languages for writing complex programs,[2] as well as an important theoretical tool.[3,4] Among other things, the ability of LISP to treat programs as data and vice versa has made it a prime choice as a host for a number of experimental languages.[5,6] However, even the most enthusiastic LISP programmers admit that the language is cumbersome in the extreme.

A couple of attempts[7,8] have been made to permit a more natural form of input language for LISP, but these are not widely available. The most ambitious of these, the LISP 2 project, bogged down in the search for efficiency.

The system described here is a less ambitious attempt to bring list-processing to the masses, as well as to create a seductive and extendable language. The name BALM is actually an acronym (Block And List Manipulator) but is also intended to imply that its use should produce a soothing effect on the worried programmer. The system has the following features.

1. An Algol-like input language, which is translated into an intermediate language prior to execution.
2. Data-objects of type list, vector and string, with a simple external representation for reading and printing and with appropriate operations.
3. The provision for changing or extending the language by the addition of new prefix or infix operators, together with macros for specifying their translation into the intermediate language.
4. Availability of a batch version and a conversational version with basic file editing facilities.

The intermediate language is actually a form of LISP 1.5 which has been extended by the incorporation of new data-types corresponding to vector, string and entry-point. The interpreter is a somewhat smoother and more general version of the LISP 1.5 interpreter, using value-cells rather than an association-list for looking up bindings, and no distinction between functional and other bindings. The system is implemented in a mixture of Fortran (!) and MLISP, a machine-

independent macro-language similar to LISP which is translated by a standard macro-assembler. New routines written in Fortran or MLISP can be added by the user, through if Fortran is used a certain amount of implementation knowledge is necessary.

The description given here is of necessity incomplete because of the flexible nature of the system. In practice it is expected that a number of different versions will evolve, with different sets of statement forms, operators, and procedures. What is described here is a fairly natural implementation of basic features of the intermediate language which will probably form the basis from which other languages will grow. We will illustrate the facilities by example rather than by giving a formal description, which can hopefully be obtained from the manual.[9]

2. Overview of BALM Features

Variables in BALM do not have a type associated with them, so each variable can be assigned any value. If we imagine the user sitting at a teletype the following conversation could ensue:

```
-A = 1.2;
-PRINT(A);
1.2
```

Lines starting with a dash are requests for the user to type. The third line is the result of the PRINT command. The usual notation is used for arithmetic operations:

```
-X = 2 * A + 3; PRINT(X);
5.4
```

with a quote symbol being used to allow the input of lists:

```
-A = "(A (B C) D);
-PRINT(HD TL A);
(B C)
```

The operators HD and TL are synonymous with CAR and CDR in LISP. Vectors can be input in a notation similar to that for lists, but using square brackets instead of parentheses. Elements of vectors are accessed by indexing:

```
-V = "[A [B C] D]; PRINT(V[2]);
[B C]
```

Lists can be members of vectors, and vice versa:

```
-PRINT(TL"(A [B C] D));
([B C] D)
-PRINT(+[A (B C) D][2])
(B C)
```

Arrays can be input as vectors of vectors, so a non-rectangular matrix could be written

```
-W = "[1 [2 3] [4 5 6]];
```

and elements can be extracted either by the usual type of indexing

```
-PRINT(W[2]);
[2 3]
```

or by repeated indexing

```
-PRINT(W[2][1]);
2
```

Assignments to vector elements are straightforward;

```
-W[2][1] = "(A B); PRINT(W[2]);
[(A B) 3]
```

A whole vector or list can be assigned from one variable to another variable in a single statement, of course, but then any operations which changes a component of one will change a component of the other. If this is not desired, the vector or list should be copied before the assignment:

```
-Z = COPY(W);
```

Character strings of arbitrary length can be specified:

```
-C = <EXAMPLE OF A STRING> ;
```

They can be concatenated, or have substrings extracted:

```
-D = C CAT  AND ANOTHER  ;
-E = SUBSTR(D,9,12); PRINT(E):
<OF A>
```

There is complete garbage collection of all inaccessible objects in the system, so the user does not need to keep track of particular lists or vectors with values of expressions as their elements, with storage being allocated dynamically:

```
-LL = LIST(X+W, "ABC, S CAT<XY>);
-VV = VECTOR)X+W, "ABC, S CAT<XY>);
```

The equivalent of the LISP CONS operator can be written as an infix colon associating to the right, so that the first of the above statements could also have been written

```
-IL = X+W: "ABC : S CAT<XY>: NIL;
```

Note that NIL is the empty list.

A procedure in BALM is simply another kind of data-object which can be assigned as the value of a variable. Machine-language procedures are specified by a name (as known to the loader) followed by 0> or 1> depending on whether they should have their arguments evaluated or not. Thus

```
-FIRST = CAR0>;
```

will assign the value of the variable FIRST as being the machine-language routine whose name is CAR. In fact, since the value of the variable CAR is also CAR0> this can also be accomplished by

```
-FIRST = NOOP CAR
```

where we have used NOOP to indicate that the subsequent operator name CAR should be regarded as a variable. Either CAR, CAR0>, or FIRST can subsequently be used to invoke this procedure.

A programmer-defined procedure is normally represented within the system in the form of a list, and is executed interpretively when invoked. The usual way of defining a procedure is to assign it as the value of a variable:

```
-SUMSQ = PROC(X,Y),X↑2+Y↑2 END;
```

The translator translates the PROC...END part into the appropriate list, which would have the same effect as if we had written

```
-SUMSQ = "(LAMBDA(X Y)(PLUS (EXPT X 2)(EXPT Y 2)));
```

Of course we could equally well have produced this list as the value of some expressions.

Instead of assigning a procedure as the value of a variable, we can simply apply it, so that

```
-X = 5 + PROC(X,Y),X↑2+Y↑2 END(2,3) + 0.5;
```

would assign 5 + 13 + 0.5 = 18.5 as the value of X. Note that procedures can accept any data-object as an argument, and can produce any data-object as its result, including vectors, lists, strings and procedures. Thus it is possible to write

```
-M = MSUM(Ml, MPROD(M2,M3));
```

where Ml, M2, M3, and M are matrices. Procedures can be recursive, of course.

Analogous to procedures we can also compute with expressions. The statement

```
-E = EXOR A + B END;
```

would assign the expression A + B, not its value, to E. Subsequently, values could be assigned to A and B, and E evaluated:

```
-A = 1; B = 2.2; V = EVAL(E);
```

EVAL(E) could also have been written as $E.

A procedure is essentially defined as an expression; for more complicated procedures, more complicated expressions can be used. The most important of these is the block, which is similar to that used in Algol, but can have a value. The statement:

```
-REVERSE = PROC(L), COMMENT ⟨REVERSES A LIST⟩
-          BEGIN(X), COMMENT ⟨X IS LOCAL VARIABLE⟩
-          COMMENT ⟨FIRST TEST FOR ATOMIC ARGUMENT⟩
-          IF ATOM(L) THEN RETURN (L),
-          COMMENT ⟨OTHERWISE ENTER REVERSING LOOP⟩
-          X = NIL,
-          COMMENT ⟨EACH TIME ROUND REMOVE ELEMENT FROM L,
-            REVERSE IT, AND PUT AT BEGINNING OF X⟩
- NXT,     IF NULL(L) THEN RETURN(X)
-          X=REVERSE(CAR (X)): COMMENT ⟨: IS INFIX CONS
-            OPERATOR⟩X,
-          L = CDR L, GO NXT,
-          END END;
```

shows the use of a block delimited by BEGIN and END in defining a procedure REVERSE which reverses a list at all levels.

As well as the IF...THEN... statement there is an IF...THEN... ELSE.... as well as an IF...THEN...ELSEIF...THEN... etc. Looping statements include a FOR...REPEAT... as well as a WHILE...REPEAT... . A label should be regarded just as a local variable whose value is the internal representation of the statements following it. Accordingly assignments to labels, and transfers to variables or expressions are legal, and can give the effect of a switch. A compound statement without local variables or transfers can be written DO...,...END. Of course any of these statements can be used as an expression, giving the appropriate value. Note that a comma is used to separate statements and labels within a block and a compound statement. The semicolon is interpreted as an end-of-command by the system (unless it occurs within a string), even if it occurs within parentheses or brackets. Any unpaired parentheses or brackets will be paired automatically, with a warning message being issued.

3. Extendability

The TRANSLATE procedure used by BALM to translate statements into the appropriate internal form is particularly simple, consisting of a precedence analysis pass followed by a macro-expansion pass. Built-in syntax is provided only for parenthesized subexpressions, comments, the quote operator ", the NOOP operator, procedure calls, and indexing. All other syntax information is provided in the form of three lists which are the values of the variables UNARYLIST, INFIXLIST, and MACROLIST. The user can manipulate these lists as he wishes, by adding, deleting, or changing operators or macros.

Operators are categorized as unary, bracket, or infix, and have precedence values, and a procedure (or macro) associated with them. Examples of unary operators are - (minus), CAR, and IF, while infix operators include +, THEN, and =. Bracket operators are similar to unary operators but require a terminating infix operator which is ignored. Examples of bracket operators are BEGIN and PROC, which both can be terminated by the infix operator END.

New operators can be defined by the procedures UNARY, BRACKET, or INFIX. These add appropriate entries onto UNARYLIST or INFIXLIST. For example the statement:

```
-UNARY("PR,150,"PRINT);
```

would establish the unary operator PR with priority 150 as being the same as the procedure PRINT. Thus we could subsequently write PR A instead of PRINT(A). Similarly we could define an infix operator → by

```
-INFIX("→,49,50,"APPEND);
```

to allow an infix append operation. The numbers 49 and 50 are the precedences of the operator when it is considered as a left-hand and right-hand operator respectively, so that an expression such as A → B → C will be analyzed as though it were A → (B → C)

The output of the precedence analysis is a tree expressed as a list in which the first element of each list or sublist is an operator or macro. For example, the statement:

```
-SQ = PROC(X), X * X END;
```

would be input as the list:

```
(SQ = PROC(X), X * X END)
```

and would be analyzed into:

```
(SETQ SQ (PROC (COMMA X (TIMES X X))))
```

This would then be expanded by the macro-expander, giving:

```
(SETQ SQ (QUOTE (LAMBDA(X) (TIMES X X))))
```

the appropriate internal form. This would then be evaluated, having the same effect as the statement:

```
(SQ = "(LAMBDA(X) (TIMES X X));
```

which would in fact be translated into the same thing.

The macro-expander is a function EXPAND which is given the syntax tree as its argument. It is actually defined as:

```
-EXPAND = PROC(TR),
-         BEGIN(Y)
-         IF ATOM(TR) THEN RETURN(TR),
-         Y = LOOKUP(CAR TR,MACROLIST),
```

```
-       IF NULL(Y) THEN RETURN (MAPCAR(EXPAND,TR)),
-       RETURN(Y(TR))
-       END END;
```

That is, if the top-level operator is a macro, it is applied to the whole tree. Otherwise EXPAND is applied to each of the subtrees recursively. Most operators will not require macros because the output of the precedence analysis is in the correct form. However, operators such as IF, THEN, FOR, PROC ... etc. require their arguments to be put in the correct form for the interpreter. For instance, the IF macro can be defined:

```
-MIF = PROC(TR),
-      BEGIN(X)
-      X = CAR CDR TR.
-      IF EQ (CAR X, "THEN) THEN RETURN
-         ("COND: LIST(EXPAND(CAR CDR X),
          EXPAND(CAR CDR CDR X)): NIL),
-      RETURN("COND: EXPAND(X))
-      END END;
```

where recursive calls to EXPAND are used to transform subtrees in the appropriate way. The statement:

```
-MACRO("IF,MIF);
```

would associate the macro MIF with the operator IF.

One particular outcome of this expansion procedure is the ability to write other than simple variables on the left-hand-side of assignment statements. There are conveniently handled by a macro associated with the assignment operator which checks the expression on the left-hand-side and modifies the syntax tree accordingly. It is this mechanism which permits an element of a vector to appear on the left-hand-side, and also such statements as:

```
-CAR(X) = Y;
```

which will be translated as though it had been written:

```
-RPLACA(X,Y);
```

The assignment macro currently in use looks up the top level operator found on the left-hand-side in the list LMACROLIST, applying any macro associated with the operator to the tree representing the assignment statement. The set of expressions which can be handled on the left-hand-side can easily be extended by adding entries to LMACROLIST. For example the procedure:

```
-LMACRO("PROP,MPROP);
```

could be used to add the left-hand-side macro MPROP to permit assignments such as:

```
-PROP("X,"P) = "V;
```

which establishes the value V for property P of atom X.

Note that the essential properties of the system are those of the intermediate language, the most important of which is its ability to treat data as program, and thus to preprocess its program. Even the TRANSLATE procedure described above can be ignored and the user's own translator substituted. Of course this will require a different level of expertise on the part of the programmer than simply the addition of

new operators. However, the current translator is only about 250 cards, and quite straightforward, so this is not an unlikely possibility.

We have not yet had much experience with the extendability features, but anticipate that we will be able to add the equivalent of the PL/1 and Algol 68 structures (as vectors with named components), and at least some of the flavor of the Snobol pattern match and substitution rule. At the very least we will have a very flexible experimental language with powerful list-processing facilities.

The translator currently takes of the order of 2000 words in the CDC 6600, and we do not expect this to increase much, if at all.

References

1. J. McCarthy et al., LISP 1.5 Programmers Manual, MIT Press, 1962.

2. M. Minsky, Semantic Information Processing, MIT Press, 1968.

3. J. Painter, Semantic Correctness of a Compiler for an Algol-like Language, A.I. Memo 44, Stanford University, 1967.

4. P. Landin, The Mechanical Evaluation of Expressions, Computer Journal, January, 1964.

5. D. Bobrow and J. Weizenbaum, List Processing and Extension of Language Facility by Embedding, IEEE Trans. on Elec. Comp. EC-13, August, 1964.

6. C. Engleman, Mathlab - A Program for On-line Machine Assistance in Symbolic Computations, Proc. FJCC, 1965.

7. D. Bobrow, D. Murphy,W. Teitelman, BBN-LISP Manual, Bolt Beranek and Newman Inc., Cambridge, Mass., April 1969.

8. P. Abrahams et al., The LISP 2 Programming Language and System, Proc. FJCC, 1966.

9. M. Harrison, BALM Users Manual, Courant Inst. Math. Sci., New York Univ. (in preparation).

(LB1, p37)

A LISP RANDOM NUMBER PACKAGE

Palme, Macob. Research Institute of National Defense, Operations Research Center. Stockholm 80, Sweden. January 1968.

1. Introduction

This is a package of routines generating pseudorandom numbers for use in the LISP programming language. Included are routines for generating random numbers with uniform and normal distribution. The package was written in LISP 1.5 for the IBM 7090 computer. I tried to write as much as possible of the routines in LISP, but three short functions (REMTI, FLOAT and IFIX) had to be written in machine code. These must thus be changed, if the package is used on another computer.

The random number series is stored in the APVAL of an atom which is specified in each call to one of the random number functions. Several different random number series can thus be used simultaneously by using different atoms in calls to the routines.

2. SETRND - setting an initial value for the generator

```
(SETRND (LAMBDA (RANDTAL RANUM)
(CSET RANUM (LOGOR RANDTAL 1))))
```

setrnd(randtal ranum)

The random number series is stored in the APVAL of ranum. This function sets an initial value to the series, using the number randtal. If randtal is even, randtal + 1 is used in the generator, because the generator requires an odd starting number.

Example of use:

```
SETRND
(1234567 RANDNUM)
```

3. REMTI - steps in the random number series

```
LAP (((REMTI SUBR 1)
(SXA RMT1 4)
(SXA RMT2 2)
(PDX 0 2)
(TSX FIXVAL 4)
(XCA)
(MPY RMT3)
(XCA)
(LDQ $OCTD)
(TSX $MKNO 4)
RMT1 (AXT 0 4)
RMT2 (AXT 0 2)
TRA 1 4)
RMT3 (1000003Q)
)NILO
```

*Specific to this LISP Assembly

Instructions:

```
(MPY 0200Q8)
(SSP 0760000000003Q)
PAC 0737Q8)
(RQL -0773Q8)
(LRS +0765Q8)
(ORA -0501Q8)
(LLS +0763Q8)
(FAD +0300Q8)
(UFA -0300Q8)
(ALS +0767Q8)
```

Entries into LISP system routines*

```
(NUMVAL +15343Q)
(FIXVAL 14065Q)
($SMKNO 13624Q)
($FLOAT 533Q)
($OCTD 540Q)
($FIXD 532Q)
```

(LB1, p38)

Random numbers are generated by the process $x_{i+1} = k \cdot x_i \pmod{2^n}$ where k is the octal number 1000003 and n is 35(decimal). REMITI takes as input x_i, and gives as output x_{i+1}.

4. EVALTIMES - useful for testing random number functions

```
(EVALTIMES (LAMBDA (FORM TIMES) (COND
((ZEROP TIMES) NIL)
(T (PROG2 (PRINT (APPLY (CAR FORM) (CDR FORM) NIL))
          (EVALTIMES FORM (SUB1 TIMES))
)))))
```

evaltimes(form times)

The argument form must be an s-expression of the kind (func arg1 arg2 ...). EVALTIMES applies func to (arg1 arg2 ...) and prints the result many times. The argument times is the number of times, that form is to be evaluated.

5. RANDLAP - gives random number as full-word integer

```
(RANDLAP (LAMBDA (RANUM)
(CSET RANUM (REMTI (VALUE RANUM)))))
(VALUE (LAMBDA (TAL) (CAR (GET TAL (QUOTE APVAL)))))
```

randlap(ranum)

This function gives a random integer, uniform in the interval 0 to $2^{35}-1$. The argument ranum is the name of the random number series to be used.

Example of use:

```
EVALTIMES
((RANDLAP RANDNUM) 10)
(1532251016250)
(2035243052770)
(1202741200750)
(1011613602670)
(2640133210450)
(3551071631570)
(725045315150)
(114330147470)
(514700466650)
(2435351644370)
```

6. FLOAT - converts an integer to a floating point number

```
LAP(((FLOAT SUBR 1)
(SXA FLT1 4)
(SXA FLT2 2)
(PDX 0 2)
(TSX FIXVAL 4)
(LRS 8)
(ORA CC2)
(STO E1)
(CLA CC3)
(LLS 8)
(FAD E1)
(LDQ $FLOAT)
(TSX $MKNO 4)
FLT11(AXT 0 4)
FLT2 (AXT 0 2)
(TRA 1 4)
E1 (0)
CC2 (243Q9)
CC3 (466Q6)
) NIL)
```

Operation codes and entries into LISP system routines are defined earlier.

7. NOLLETTRAND - gives a uniform floating random number between 0 and 1

```
(NOLLETTRAND (LAMBDA (RANUM)
(TIMES (FLOAT (CAR (RANDLAP RANUM)))
       0.29103830E-10)))
```

nollettrand(ranum)

This function gives a random number with uniform distribution in the interval between 0 and 1. The argument ranum is the name of the random number series to be used.

Example of use:

```
EVALTIMES
((NOLLETTRAND RANDNUM) 10)
0.4191065E0
0.51431566E0
0.31393551E0
0.25477194E0
0.70321202E0
0.92632463E0
0.22903957E0
0.3731556E-1
0.16253719E0
0.63938313E0
```

8. FLOATRAND - gives a uniform floating random number in a given interval

```
(FLOATRAND (LAMBDA (FLOATMIN FLOATMAX RANUM)
(PLUS FLOATMIN
      (TIMES (DIFFERENCE FLOATMAX FLOATMIN)
      (NOLLETTRAND RANUM)))))
```

floatrand(floatmin floatmax ranum)

This function gives a random number with uniform distribution in the interval between floatmin and floatmax. The argument ranum is the name of the random number series to be used.

9. RANDSUM - gives the sum of a given number of uniform random numbers

```
(RANDSUM (LAMBDA (INT RANUM)
(COND ((ZEROP INT) 0.)
      (T (PLUS (NOLLETTRAND RANUM)
               (RANDSUM (SUBl INT) RANUM))))))
```

randsum(int ranum)

This function computes int different random numbers, all uniform in the interval between 0 and 1. The value of the function is the sum of the int numbers.

10. RNORM - gives a floating random number with normal distribution

```
(RNORM (LAMBDA (MEAN DEVIATION RANUM)
(PLUS MEAN
      (TIMES DEVIATION
             (PLUS -6.0 (RANDSUM 12 RANUM))))))
```

rnorm(mean deviation ranum)

This function computes a random number with normal(Gaussian) distribution. mean is the mean value of the distribution, deviation is its standard deviation.

Example of use:

```
EVALTIMES
((RNORM 0.0 0.1E1 RANDNUM) 10)
-0.94025826EO
0.15263649E1
-0.19060186E1
-0.9052363EO
0.10646668E1
-0.81394666EO
0.3982892EO
0.48880923EO
-0.52285045EO
-0.13955505E1
```

11. IFIX - converts a floating number to an integer

```
LAP(((IFIX SUBR 1)
(SXA FLT1 4)
(PDX 0 4)
(CLA 0 4)
(PDX 0 4)
(CLA 0 4)
(UFA CC4)
(RQL 8)
(LRS 8)
(SCA)
(FLT1 (AXT 0 4)
(LDQ $FIXD)
(TRA $MKNO)
CC4 (266Q9)
) NIL)
```

Operations and entries to the LISP system routines are defined earlier.

IFIX truncates the floating number. Truncation is done towards zero, that is, positive numbers are truncated to the nearest smaller integer, and negative numbers to the nearest larger integer.

12. FIXRAND - gives a uniform integer random number

```
(FIXRAND (LAMBDA (FIXMIN FIXMAX RANUM)
(PLUS FIXMIN
      (IFIX (TIMES (ADD1 (DIFFERENCE FIXMAX FIXMIN))
                   (NOLLETTRAND RANUM))))))
```

fixrand(fixmin firmax ranum)

This function computes a random integer in the interval between fixmin and fixmax including both ends of the interval. The integer is uniformly distributed. fixmin and fixmax must be integers.

Example of use:

```
EVALTIMES
((FIXRAND 1 10 RANDNUM) 10)
5
6  etc.
```

Daniel G. Bobrow, Editor
Lisp Bulletin
Computer Science Division
Bolt Beranek and Newman Inc.
50 Moulton Street
Cambridge, Massachusetts 02138
telephone: 617/491-1850